

## Description

`frames use` loads into memory a set of frames from a Stata frameset (`.dtas`) file previously saved by `frames save`.

## Quick start

Load all frames in file `myframeset.dtas`

```
frames use myframeset
```

Load frames A and B in file `myframeset.dtas`

```
frames use myframeset, frames(A B)
```

## Menu

Data > Frames Manager

## Syntax

```
frames use filename [ , options ]
```

If *filename* is specified without an extension, `.dtas` is assumed. If *filename* contains embedded spaces or other special characters, it has to be enclosed in double quotes.

<i>options</i>	Description
<code>frames(<i>framelist</i>)</code>	specify frames to be used
<code>clear</code>	clear all frames in memory and replace them with the frames from disk
<code>replace</code>	overwrite existing frames in memory with frames of the same name from <i>filename</i>

## Options

`frames(framelist)` specifies the frames to be loaded into memory. *framelist* is a list of frame names separated by a space. If `frames()` is not specified, all frames are loaded. The `frames()` option does not change the current working frame; to change the working frame after `frames use`, use `frame change`.

`clear` clears all frames in memory and replaces them with frames from disk. The new working frame will be the first frame that was specified in the `frames(framelist)` option with `frames save`.

If both `clear` and `frames(framelist)` are specified with `frames use`, the new working frame will be the first one listed in *framelist*.

`replace` replaces frames in memory with frames from *filename* if the frame names are the same. This option does not drop from memory existing frames with different names.

## Remarks and examples

`frames use` is used to load a frameset previously saved with `frames save`. A frameset is a single file with data from multiple frames. By loading a frameset with `frames use`, you can resume the work you were doing with the frames saved with `frames save`.

`frames use` will load all the data frames stored in the `.dtas` file, unless you specify a list with the `frames()` option. Additionally, when no other options are specified, the frames will be loaded into memory, but the current working frame will not be changed, even if it is empty. When `frames use` is specified with both the `frames()` and `clear` options, the new working frame will be the first frame listed in the `frames()` option. When `frames use` is specified with the `clear` option but without the `frames()` option, the new working frame will be the first frame that was specified in the `frames()` option with `frames save`. Note that the first frame of a `.dtas` file is stored in `r(first)` after the frameset is described with `frames describe using`.

### ► Example 1

To demonstrate how to load a frameset, we first need to create a frameset. Below, we create frames `census` and `housing` with data from the 1980 census. We then use `frames save` to store both of these frames in a file named `myframeset.dtas`.

```
. clear frames
. sysuse census
(1980 Census data by state)
. frame rename default census
. frame create housing
. frame change housing
. webuse hsnrg
(1980 Census housing data)
. frames save myframeset, frames(census housing) replace
(file myframeset.dtas not found)
file myframeset.dtas saved
```

Suppose that at a later time we would like to load the frames in `myframeset.dtas`. We first clear any data and frames and then use `frames use`.

```
. clear all
. frames use myframeset
  census   50 x 13; 1980 Census data by state
  housing  50 x 12; 1980 Census housing data
```

We see in the output above that both frames were loaded into memory. If there is no dataset in memory, `frames use` loads the frames from the `.dtas` file, but the default frame remains the current working frame, as shown below:

```
. pwf
(current frame is default)
```

The output from `pwf` shows that the current frame is `default`. `frames describe` below lists the frames in alphabetical order and shows that the default frame (labeled `Frame: default`) is empty:

```
. frames describe
```

Frame: census				
Contains data				
Observations:	50	1980 Census data by state		
Variables:	13	28 Mar 2025 19:42		
Variable name	Storage type	Display format	Value label	Variable label
state	str14	%-14s	cenreg	State
state2	str2	%-2s		Two-letter state abbreviation
region	int	%-8.0g		Census region
pop	long	%12.0gc		Population
poplt5	long	%12.0gc		Pop, < 5 year
pop5_17	long	%12.0gc		Pop, 5 to 17 years
pop18p	long	%12.0gc		Pop, 18 and older
pop65p	long	%12.0gc		Pop, 65 and older
popurban	long	%12.0gc		Urban population
medage	float	%9.2f		Median age
death	long	%12.0gc		Number of deaths
marriage	long	%12.0gc		Number of marriages
divorce	long	%12.0gc		Number of divorces
Sorted by:				
Frame: default				
Contains data				
Observations:	0			
Variables:	0			
Sorted by:				

Frame: housing				
Contains data				
Observations:	50			1980 Census housing data
Variables:	12			28 Mar 2025 19:42
Variable name	Storage type	Display format	Value label	Variable label
state	str14	%14s		State
division	int	%8.0g	division	Census division
region	int	%8.0g	region	Census region
pop	long	%10.0g		Population in 1980
popgrow	float	%6.1f		Pop. growth 1970-80
popden	int	%6.1f		Pop/sq. mile
pcturban	float	%8.1f		Percent urban
faminc	long	%8.2f		Median family inc., 1979
hsng	long	%10.0g		Hsng units 1980
hsnggrow	float	%8.1f		% housing growth
hsngval	long	%9.2f		Median hsng value
rent	long	%6.2f		Median gross rent
Sorted by: state				

If there are frames in memory, frames census and housing in myframeset.dtas will be loaded into memory, in addition to the frames already in memory. If there is already a frame in memory with the same name as the frame you are loading, frames use issues an error message. For example, below we rename the default frame to census and then run our frames use command once more:

```
. clear frames
. sysuse census, clear
(1980 Census data by state)

. frame rename default census
. frames use myframeset.dtas
frames in memory are in conflict with frames on disk
  Frame census is already in memory. Specify option clear to clear all
  frames or option replace to replace only the frames in conflict.
r(4);
```

To successfully load the frames from myframeset.dtas, we can either use the clear option to clear all frames from memory,

```
. frames use myframeset, clear
```

or use the replace option to replace the frames in conflict:

```
. frames use myframeset, replace
```



Stored results

frames use stores the following in r():

```
Macros
  r(fn)           pathname of frameset
  r(frames)       list of frames loaded
```

## Also see

[D] **frames describe** — Describe frames in memory or in a file

[D] **frames modify** — Modify a set of frames on disk

[D] **frames save** — Save a set of frames on disk

[D] **frames** — Data frames

[D] **sysuse** — Use shipped dataset

[D] **use** — Load Stata dataset

[D] **webuse** — Use dataset from Stata website

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).